
Outline: 

1. Manually generate market shares & within group market shares 
2. Run logit & nested-logit estimation 
3. Manually calculate marginal cost & price-cost margins under Bertrand competition 
4. Automatically generate marginal cost & price-cost margins   
5. A simple counterfactual example of a hypothetical merger of two firms  

 

 Step 0: Loading the dataset verboven_cars.dta 
 
 Step 1: Preparation for demand estimation: manually generate market shares 
 

1) Generate new variables gen logq = ln(qu) 
 

gen logp = ln(pr) 
gen logpop = ln(pop) 
gen loggdp = ln(ngdp) 
gen msize = pop/4 

 
2) Construct market share s_j 

gen share = qu/msize 
egen sum_share = sum(share), by(ma ye) 

 
3) Construct outside good's market share s_0 
 

gen share0 = 1 - sum_share 
 

sum share share0 
 

4) Generate log odd ratio (all above have done in tutorial 3) 
 

gen lsj_ls0 = ln(share/share0) 
 

5) Manually generate market shares & within group market shares 



Groups are defined as whether the car is a domestic or foreign car, which is denoted as variable 
“home” in the dataset. This is a simpler version of Nested -Logit of Bjornerstedt & Verboven 
Stata J.pdf.  

sum home 

 

If home=1, it means the car is domestic; home=0 represents a foreign car. 

The following code is generating the sum of all domestic products’ market shares in a given 
year and country and the sum of all foreign products’ market shares in a given year and 
country. 

bys ma ye home: egen denom= total(share) 

For a given product, the percentage it accounts for among all domestic/foreign cars is the 
within-group market share: 

gen s_within = share/denom 

Then we generate its logarithm form. 

gen ln_s_within=ln(s_within) 

 
   we generate the denominator of within-group market shares: for a given country and year, the 

summation of all products’ market shares should contain at most 2 values: one for domestic 
group; the other for foreign group. We can check by using the following commands: 

 
 

bys ye ma: tab denom 



 
 
 

Step 2: Logit & Nested-logit estimation 

1) Logit estimation including model attributes:  
 

reghdfe lsj_ls0 logp hp li wi cy le he logpop loggdp, vce(robust) a(ma ye brd) 

	  



 
2) Nested-logit estimation including model attributes: 

 
The reason why we would like to try Nested-logit estimation is that the standard Logit 
model imposes strong restrictions on price elasticities (the IIA property). 
By simply adding one more explanatory variable: the logarithm of within-group market 
shares into the regression, the nested-logit model is estimated. 

 
reghdfe lsj_ls0 logp ln_s_within hp li wi cy le he logpop loggdp, vce(robust) a(ma ye brd) 

 
 
Comments: Nested-logit is not valid here since the coefficient of within-group market shares 
is larger than 1. We could reconsider the nests by setting groups as cars’ sizes: compact, suv, 
mpv, etc. then setting subgroups as domestic & foreign cars. This type of nests corresponds to 
the paper: Bjornerstedt & Verboven Stata J.pdf. 
 
 

Step 3: Manually calculate marginal cost & price-cost margins under Bertrand competition 

 

1) Rerun the Logit model 



Since Nested logit does not perform well here, we use standard logit to calculate marginal 
cost and price-cost margins. Before calculating the price-cost margin, we need to rerun the 
standard logit estimation to let Stata memorize the coefficients. 
 
reghdfe lsj_ls0 logp hp li wi cy le he logpop loggdp, vce(robust) a(ma ye brd) 
 
2) Generating margins 

 
The coefficient of price is saved in Stata’s memory as _b[logp]. We can calculate the price-
cost margin as follows. Notice that this equation is for the Bertrand equilibrium, and the 
corresponding measure of the margin and the marginal revenue, is based on the assumption 
that each product is produced by only one firm, or that a multiproduct firm maximizes the 
profit from each product separately, without taking into account cannibalization effects. 
Additionally, the price enters in the utility as its logarithm form. That’s why the equation 
needs to be multiplied by the price on the top. 
 
gen margin=-eurpr/(_b[logp]*(1-share)) 
 
Here the margin is measured in monetary units, in euros.  
 
sum margin 

 
 
3) Generating marginal revenue  

 
Now, we can calculate the marginal revenue by subtracting the margin from price. 
 
gen mr= eurpr-margin 
gen log_mr= ln(mr) 

 

 
4) Estimating marginal costs 



 
By running an OLS regression on marginal revenue using the quantities and product’s 
attributes as explanatory variables, we could estimate marginal cost as follows: 
 
reghdfe log_mr logq hp li wi cy le he logpop loggdp, vce(robust) a(ma ye brd)  
 
 

 
 
 
There is an endogeneity problem, i.e., logq is negatively correlated with the unobserved 
component of the cost, which could provide a downward biased estimate of the true 
coefficient of logq. As usual, we need to find IV for logq. However, to make our life easier 
here, we could impose the assumption of constant marginal costs as follows: 
 
reghdfe log_mr hp li wi cy le he logpop loggdp, vce(robust) a(ma ye brd) res 
 



 
 
predict log_mc_hat, xbd 
 
gen mc= exp(log_mc_hat) 
 
We predict marginal cost by including all FEs using “xbd”. Then marginal cost is generated 
by taking the exponential form of the prediction. 
 
 

 

 
 
 

 
 
5) Taking the average marginal cost for each firm cross years 

 
bys frm : egen mc_frm=mean(mc) 
 



or we could take the average marginal cost for each firm in a given year: 
 
bys frm ye: egen mc_frm=mean(mc) 
 
 

Step 4: Automatically generate marginal cost & price-cost margins   

Note: here, step 1 (1) is the only necessary part for generating marginal cost & price-cost 
margins automatically. Secondly, using command “mergersim” gives a different estimate 
of margins, mr, and mc is because that the command takes into account that these firms 
have multiproduct, and therefore the expression for the f.o.c is not same as the previous 
mc’s manual calculation. The previous estimation, by contrast, is assuming each firm has 
only one product. Additionally, “mergersim” sets price in levels as an explanatory 
variable rather than the logarithm form that we use in the previous manual estimation. 
 
1) Set the dataset as 2 dimensions of the panel data: the car model and the market (year & 

country) 
 

egen yect=group(ye ma),label 
xtset co yect 
 
2) Initializing the merger settings  

 
 
If you want to use nested-logit, do the following. The code will generate a group share variable 
“M_lsjg” into the dataset if setting the nests. The market shares are also generated automatically, 
which is denoted as M_ls. There is no need to generate market shares manually. 
 
mergersim init,nests(home) price(eurpr) quantity(qu) marketsize(msize) firm(frm) 

 
 



If using standard logit, we could use following code: 
 
mergersim init, price(eurpr) quantity(qu) marketsize(msize) firm(frm) 
 

 
 

3) Set the dependent variable in demand estimation as M_ls. 
 
Stata needs to know which regression you would like to use and the dependent variable should 
be same as the dependent variable that is shown in the above table. Note: here price doesn’t take 
the logarithm form since it should be consistent with the setting of part (2)  
 

reghdfe M_ls eurpr hp li wi cy le he logpop loggdp, vce(robust) a(ma ye brd) 
 

 



 
 
4) 	 Show firm's marginal cost & pre-merger Lerner 

 
If you want to look at average mc for each firm cross years: 
 

mergersim market 
 
If you want to see the average mc for each firm in a given year, for example, 1983: 
 
mergersim market if ye == 83 
 

 
 
 
Step 5: A simple counterfactual example of a hypothetical merger of two firms 
 
1) check market & firm's label 

 
d 

 
 
By looking at the descriptions of variables, we know the label of firm is “firm” and label of market 



is just “market”. Then we would like to see the label list for “firm” & “label”. 
 
label list firm 
label list market 
 
 

 

 

 
2) The example below considers a merger where General Motors (GM) (.#15) sells its operations 

to VW (#26) for German in 1998 
 
mergersim simulate if ye == 98 & ma == 3, seller(15) buyer(26) detail 
 
 
The following tables show the price change after merger between GM & VW and pre & post 
merger’s market shares. 
 



 
 



 
 
 
 
3) Plot the graph of price change 
 
gen perc_price_ch=M_price_ch*100 
 
graph bar (mean) perc_price_ch if ma==3 & ye==98, /// 
over(frm, sort(perc_price_ch) descending label(angle(vertical))) /// 



ytitle(Percentage) title(Average percentage price increase per firm) 
 

 


